Quantum Wells −− The Shooting Method (application to the simple harmonic oscillator and other problems of a particle in a potential)

Introduction

In the previous handout we found the eigenvalues of a quantum particle in a potential well where the potential vanishes for |x| greater than some value, which has the advantage that we know the wavefunction exactly in this large |x| region. The same method can be used for problems where the potential, while not exactly zero at large |x|, is sufficiently close to zero that the error in assuming it vanishes is negligible.

Here we give a generalization of this approach to problems where the potential does *not* tend to zero at large |x|. This more general approach is often called the *shooting method* . This handout is very similar to the earlier one except for the way it handles the boundary conditions at large |x|.

Setting up the Problem of the Simple Harmonic Oscillator

As an illustration,we take the **simple harmonic oscillator** (SHO) potential with $\hbar=\omega=m=1$,for which **there is an analytic solution**, discussed in all books on quantum mechanics. The energy levels are

$$E_n = n + \frac{1}{2} \quad , \quad (n = 0,\ 1,\ 2,\ \ldots)$$

First we set up the potential and plot it. The potential below is acually of the more general form

$$V(x) = V_0 \frac{|x|^k}{k}$$

where k can be arbitrary, but here we set

$$k = 2$$

the SHO. However, simply by changing k one can solve a range of problems using this notebook.We shall study k = 4 at the end.

```
In[1]:= Clear["Global‘*"]

In[2]:= k = 2;

In[3]:= v0 = 1;

In[4]:= v[x_] := v0 Abs[x]^k / k
```

```
In[5]:= Plot[v[x], {x, -4, 4},
          AxesStyle -> AbsoluteThickness[1], AxesLabel -> {"x", "V(x)"}];
```



We define the Schrodinger equation using a delayed assignment, ":=", since we will only use it later:

```
In[6]:= eqn[en_] := u''[x] + 2 (en - v[x]) u[x]
```

(we call the wavefunction u(x) here.) We will solve the equation in the range $-L \le x \le 0$, and choose L such that V(x) >> E at x=−L, i.e. x = −L is well to the left of the "turning point" where V(x) = E. We take L = 4, which is fine for the lowest level. However we will need to increase L to get the higher levels accurately.

```
In[7]:= L = 4;
```

At x = −L, we take, arbitrarily, u(−L) = 0, u'(−L) = 1. This corresponds to a *linear superposition* of the solution which decays exponentially to the left and the one which increases exponentially. We just want the solution which decreases exponentially. However, if L is deep inside the region where V(x) > E the error will be negligible since we integrate to the right and so the unwanted solution will be exponentially surpressed as we integrate towards the negative−x turning point.

We set up the calculation of the wavefunction in the region between −L and 0, matching the function and its derivative to the specified values at x = −L.

```
In[8]:= wavefunc[en_] := NDSolve[{eqn[en] == 0, u[-L] == 0,
                          u'[-L] == 1}, u, {x, -L, 0}]
```

Note that "**wavefunc[en_]**" will given as a replacement rule in the form "**{{u→InterpolatingFunction[{{−0.5,0.5}},"<>"]}}**". In order to directly access the wavefunction we define a function, called **sol[x, en]**, which applies the replacement rule, and removes one of the sets of curly brackets by taking the first element of the list.

```
In[9]:= sol[x_?NumericQ, en_?NumericQ] := u[x] /. wavefunc[en][[1]]
```

Note that we have added the hieroglyphics

?NumericQ

after the arguments of **sol**. This is necessary in *Mathematica* 5 when the solution is put into **FindRoot** below to determine the energy eigenvalue. (**?NumericQ** imposes that the function is only evaluated if the arguments are real.)

It is also convenient to define a function for the derivative of the wave function

(since we will be requiring that this is zero at x = 0 for the even parity solutions):

```
In[10]:= solprime[x_?NumericQ, en_?NumericQ] := u'[x] /. wavefunc[en][[1]]
```

Even Parity Solution

We now find an eigenvalue. We use the "FindRoot" command to locate the eigenvalue and give it two starting values. The boundary condition is that the derivative of the wavefunction is zero at x = 0:

```
In[11]:= evalue = en /. FindRoot [ solprime[0, en] , {en, 0, 1} ]

Out[11]= 0.5
```

This agrees with the known ground state energy of the simple harmonic oscillator, $E_0 = 1/2$.

Now we want the eigenfunction coresponding to our eigenvalue. Since we now have the eigenvalue, we do not want to keep recalculating the wavefunction so we define a function "**efunc**" with immediate assignment, where we input the eigenvalue for the energy:

```
In[12]:= efunc[x_] = u[x] /. wavefunc[evalue][[1]];
```

We have now obtained the wavefunction in for x > 0. We now also define it for x < 0 (remembering that it's even) and collect these functions into a single (not normalized) function ψnn[x_] , which can then easily be plotted:

```
In[13]:= ψnn[x_] := efunc[x] /; x ≤ 0 ;

In[14]:= ψnn [x_] := efunc[-x] /; x > 0 ;
```

We now normalize the wavefunction,

```
In[15]:= normconst = Sqrt[ NIntegrate [ ψnn[x]^2, {x, -L, L} ] ];

In[16]:= ψ[x_] := ψnn[x] / normconst;
```

and then plot it:

```
In[17]:= fig = Plot[ψ[x], {x, -L, L}, AxesLabel -> {"x", "ψ"},
           PlotRange → {0, 1}, DisplayFunction -> Identity ];
```

```
In[18]:= Show[fig, Graphics[{
                  Text[evalue, {1.6, 0.9}, {-1, 0}],
         Text["E = ", {1.6, 0.9}, {1, 0}],
                  Text[v0, {2.6, 0.6}, {-1, 0}],
         Text["V0 = ", {2.6, 0.6}, {1, 0}]}], DisplayFunction -> $DisplayFunction];
```



We see that there are no nodes (zeroes) in the wavefunction which means, since we are in one dimension, that it is the ground state.

Odd Parity Solution

Now we look at odd-parity solutions.

We repeat the previous calculation of the eigenvalue, and calculate the eigenfunction, which is then normalized and plotted. We give different initial guesses for the eigenvalue from what we took for the even parity solution and also take a somewhat larger value for L, in order to get an accurate answer for this state which has higher energy than the even parity solution discussed in the previous section.

```
In[19]:= L = 5;
```

The boundary condition is now that the wavefunction vanishes at the origin:

```
In[20]:= evalue = en /. FindRoot[sol[0, en], {en, 1, 3}]
```

```
Out[20]= 1.5
```

This agrees with the known energy of the first excited state of the simple harmonic oscillator, $E_1 = 3/2$.

Next we calculate the wavefunction, normalize it and plot it

```
In[21]:= efunc[x_] = u[x] /. wavefunc[evalue][[1]];
```

```
In[22]:= ψnn[x_] := -efunc[-x] /; x > 0
```

```
In[23]:= normconst = Sqrt[NIntegrate[ψnn[x]^2, {x, -L, L}]];
```

```
In[24]:= fig = Plot[ψ[x], {x, -L, L}, AxesLabel -> {"x", "ψ"}, DisplayFunction -> Identity];

In[25]:= Show[fig, Graphics[{
                  Text[evalue, {1.6, 0.4}, {-1, 0}],
         Text["E = ", {1.6, 0.4}, {1, 0}],
                  Text[v0, {1.6, 0.6}, {-1, 0}],
         Text["V0 = ", {1.6, 0.6}, {1, 0}]}], DisplayFunction -> $DisplayFunction];
```



The wavefunction is smooth and has only one node, showing that it is the lowest energy odd−parity eigenstate.

The "shooting method" described in this handout can be applied to essentially any quantum well problem in one dimension with a symmetric potential. The main thing is to ensure that L is far enough into the region where the solution is exponentially decaying that the boundary conditions applied at x = −L do not introduce a noticeable amount of the "wrong" solution in the x−region of interest.

Anharmonic Oscillator

Now we consider a problem **for which there is no analytic solution**; an oscillator with a *quartic* potential:

```
In[26]:= k = 4; L = 4;
```

$$V(x) = \frac{x^4}{4}$$

(Note that in this model there is **no** x^2 term in the potential.) We look for the lowest eigenvalue, for which the eigenfunction will be **even**.

```
In[27]:= evalue = en /. FindRoot[solprime[0, en], {en, 0, 1}]

Out[27]= 0.420805
```

```
In[28]:= efunc[x_] = u[x] /. wavefunc[evalue][[1]];

In[29]:= ψnn[x_] := efunc[-x] /; x > 0

In[30]:= normconst = Sqrt[ NIntegrate[ψnn[x]^2, {x, -L, L}]];

In[31]:= fig = Plot[ψ[x], {x, -L, L}, AxesLabel -> {"x", "ψ"}, DisplayFunction -> Identity];

In[32]:= Show[fig, Graphics[{
              Text[evalue, {1.6, 0.4}, {-1, 0}],
         Text["E = ", {1.6, 0.4}, {1, 0}],
              Text[ v0, {1.6, 0.6}, {-1, 0}],
         Text["V0 = ", {1.6, 0.6}, {1, 0}]}], DisplayFunction -> $DisplayFunction];
```



Hence the lowest eigenvalue is 0.420809, which compares with 0.5 for the quadratic potential.

The **first excited even** eigenstate can be obtained from

```
In[33]:= evalue = en /. FindRoot[solprime[0, en], {en, 2.5, 3.5}]

Out[33]= 2.9588

In[34]:= efunc[x_] = u[x] /. wavefunc[evalue][[1]];

In[35]:= normconst = Sqrt[ NIntegrate[ψnn[x]^2, {x, -L, L}]];
```

and then plotted

```
In[36]:= fig = Plot[ψ[x], {x, -L, L}, AxesLabel -> {"x", "ψ"}, DisplayFunction -> Identity];
```

```
In[37]:= Show[fig, Graphics[{
                  Text[evalue, {1.6, 0.4}, {-1, 0}],
         Text["E = ", {1.6, 0.4}, {1, 0}],
                  Text[v0, {1.6, 0.6}, {-1, 0}],
         Text["V0 = ", {1.6, 0.6}, {1, 0}]}], DisplayFunction -> $DisplayFunction];
```



As expected there are two nodes.

We can also get the lowest **odd** eigenstate:

```
In[38]:= evalue = en /. FindRoot[sol[0, en], {en, 1, 2}]
Out[38]= 1.5079

In[39]:= efunc[x_] = u[x] /. wavefunc[evalue][[1]];

In[40]:= ψnn[x_] := -efunc[-x] /; x > 0

In[41]:= normconst = Sqrt[NIntegrate[ψnn[x]^2, {x, -L, L}]];

In[42]:= fig = Plot[ψ[x], {x, -L, L}, AxesLabel -> {"x", "ψ"}, DisplayFunction -> Identity];
```

```
In[43]:= Show[fig, Graphics[{
                  Text[evalue, {1.6, 0.4}, {-1, 0}],
         Text["E = ", {1.6, 0.4}, {1, 0}],
                  Text[v0, {1.6, 0.6}, {-1, 0}],
         Text["V0 = ", {1.6, 0.6}, {1, 0}]}], DisplayFunction -> $DisplayFunction];
```



We see that unlike the simple harmonic oscillator, the energy levels, 0.420805, 1.5079, 2.9588, ... , are not evenly spaced.